


WHO DUNNIT



Looking through the window you see him standing in his study. Then you hear a gun shot and he falls to the ground. You walk into the house and go into his study.

There are four men there. You know their names. You find a note which the victim must have written as he died. This is a clue to his murderer. You must decide which of the four men is the murderer before they slip out of the room.

How to play

The victim's note is by the man lying down. You must work out which of the names of the other men has some connection with this word. For example, in the screen show above, Mr Lager is the murderer as Lager and

Cider are both drinks. Alternatively words that are related may have the same or opposite meanings. For example, Big and Large, also Hot and Cold are related.

Key in the number below the suspected murderer before the fourth man disappears off the screen.

If you're right, you hear police sirens as their cars approach. If you're wrong or too late, you don't. The score is given on the top line. Press RETURN to play again.

Programming hints

This program illustrates how the ENVELOPE statement can be used to create sound effects. This allows the pitch of the amplitude (volume) of a note to be continuously varied in a short time or both. As a gun shot has a constant pitch the three pitch parameters of the ENVELOPE statement in line 410 are zero. As a gun shot has a sharp attack phase and a slower delay the volume initially shoots up to a maximum 126 before going down more slowly. This is the 126 after the 10. ENVELOPE statements are quite tricky but you can find the ENVELOPE statement you require with limited knowledge by trial and error.

If you want to add more words to the game, add some more DATA statements at the end of the program. Put in sets of three words that are related. Read the other words in lines 740 and 760 for ideas. Make sure that each set is not related to the other words in those lines. When you have added the extra words, count up the total number of sets of words from line 740 and assign it to TT in line 310.

```

10 REM WHODUNNIT
20 REM COPYRIGHT (C) G.LUDINSKI 1983
30 MODE 5
40 DIM WD$(30,3),N(5),X(4),Y(4),P1$(2
),P2$(2)
50 X(1)=0:Y(1)=7:X(2)=10:Y(2)=7:X(3)=
0:Y(3)=14:X(4)=10:Y(4)=14
60 SC=0:TU=0
70 GOTO 200
80 REM
90 REM U.D.G. CALCULATOR
100 REM
110 DEF FNB(N$)
120 TF=0
130 FOR L=0 TO 7
140 TF=TF+(2^L)*VAL(MID$(N$,8-L,1))
150 NEXT L
160 =TF
170 REM
180 REM SHAPES OF MEN
190 REM
200 BD=FNB("001111100"):LG=FNB("001110
00")
210 VDU 23,224,FNB("00010000"),FNB("00
111000"),FNB("00111000"),FNB("00010000")
,BD,BD,BD,BD
220 VDU 23,225,BD,BD,LG,LG,LG,LG,LG,LG
230 VDU 23,226,0,0,0,0,FNB("00001111")
,FNB("01011111"),255,255
240 VDU 23,227,0,0,0,0,FNB("11000001")
,255,255,255
250 M$=CHR$(224)+CHR$(10)+CHR$(8)+CHR$
(225):V$=CHR$(226)+CHR$(227)
260 B=128:BL=0:RD=1:YE=2:WH=3
270 COLOUR B+BL:COLOUR WH
280 REM
290 REM READ WORDS
300 REM
310 TT=25
320 FOR II=1 TO TT
330 READ WD$(II,1),WD$(II,2),WD$(II,
3)
340 NEXT II
350 TIME=0
360 REM
370 REM UPRIGHT MAN AND GUN SHOT
380 REM
390 CLS:COLOUR B+BL:COLOUR WH:PRINT TA
B(0,20);M$
400 *FX15,0
410 ENVELOPE1,1,0,0,0,100,20,10,126,-5
0,-1,-76,126,126:SOUND0,1,6,255
420 FOR I=1TO4
430 N(I)=RND(TT)
440 NEXT I
450 IF N(1)=N(2) OR N(1)=N(3) OR N(1)=
N(4) OR N(2)=N(3) OR N(2)=N(4) OR N(3)=N
(4) THEN GOTO390
460 REM
470 REM DRAW PEOPLE AND NAMES
480 REM
490 TU=TU+1
500 SN=RND(4)
510 FOR I=1TO2
520 P1$(I)=WD$(N((2*I)-1),RND(3)):P2
$(I)=WD$(N(2*I),RND(3))
530 COLOUR WH:PRINTTAB(0,7*I);M$;;CO
LOUR YE:PRINT" Mr ";P1$(I);:COLOUR WH:PR
INTTAB(10,7*I);M$;;COLOUR YE:PRINT" Mr "
;P2$(I)

```

```

540 PRINTTAB(0,(7*I)+3);2*I-1;TAB(10
,(7*I)+3);2*I
550 NEXT I
560 VC$=WD$(N(SN),RND(3)):IF VC$=P1$(1
) OR VC$=P2$(1) OR VC$=P1$(2) OR VC$=P2$(
2) THEN 560
570 MU$=" "+VC$+" "
580 PRINTTAB(0,21)" "
590 COLOUR WH:PRINTTAB(0,21);V$;:COLOU
R B+WH:COLOUR WH:PRINT TAB(10,20);MU$;:C
OLOUR RD:PRINT TAB(10,21);MU$;:COLOUR WH
:PRINT TAB(10,22);MU$:COLOUR B+BL
600 K=1
610 I$=INKEY$(100):IF I$="" AND K<5 TH
EN PRINTTAB(X(K),Y(K))" "TAB(X(K),Y(K)+1
)" " :K=K+1:GOTO610
620 REM
630 REM POLICE SIREN
640 REM
650 IF VAL(I$)=SN THEN FOR K=1 TO 4:SO
UND 1,-15,109,10:SOUND 1,-15,101,10:NEXT
:SC=SC+1
660 REM
670 REM SCORE
680 REM
690 COLOUR WH:PRINTTAB(0,1);SC;" solve
d"TAB(10,1);TU;:IF TU > 1 THEN PRINT " m
urders" ELSE PRINT " murder "
700 COLOUR WH:PRINT TAB(0,28)"Press Re
turn":INPUT RG$:GOTO 390
710 REM
720 REM DATA
730 REM
740 DATA Big,Small,Large,Fat,Thin,Plum
p,Quiet,Loud,Noisy,Wet,Dry,Damp
750 DATA Hot,Cold,Warm,A,Z,Alpha,Good,
Bad,Nice,Mad,Crazy,Sane,Dull,Shiny,Matt,
See,Hear,Feel,Old,Young,Aged,Laugh,Cry,W
eep,Kid,Child,Adult,Am,Pm,Noon,Bird,Fowl
,Beast,Snow,Ice,Sleet,Beer,Lager,Cider
760 DATA King,Queen,Jack,Give,Take,Gra
sp,BBC,ITV,Ch. 4,Ill,Well,Sick,Gym,PT,PE
,Red,Amber,Green,Load,Save,Chain,Eye,I,A
ye

```